

Lab6: Tables and Figures

Introduction to Econometrics, Spring 2023

Jiayi Cheng

Nanjing University

3/24/2023

Section 1

Basic Graph

Subsection 1

Introduction

Basic Graph : Introduction

- 1. Stata 图形的种类

- ▶ graph twoway 二维图

| | |
|-----------|-------|
| scatter | 散点图 |
| line | 折线图 |
| lfit | 线性拟合图 |
| histogram | 直方图 |
| kdensity | 密度函数图 |
| function | 函数图 |

- ▶ graph bar 条形图

- ▶ graph box 箱形图

- ▶ graph pie 饼图

- ▶

Basic Graph : Introduction

- 2. 二维图命令

- ▶ 语法

`tway` (单元图 1) (单元图 2) (...), 选项 1 选项 2 ...

`tway` 单元图 1 || 单元图 2 || ..., 选项 1 选项 2 ...

单元图的定义: (单元图类型 $y_1 y_2 \dots x$, 选项 1 选项 2 ...)

选项的定义: (定义内容, 子选项 1 子选项 2 ...)

- ▶ 注意: 逗号后全部为选项, 裸露在外的逗号只有一个

Basic Graph : Introduction

● 2. 二维图命令

▶ 实例 1

```
. sysuse sp500, clear
(S&P 500)

. twoway (line high date) (line low date),           ///
        title("图1: 股票最高价与最低价时序图", box)      ///
        xttitle("交易日期", margin(medsmall))           ///
        ytitle("股票价格")                               ///
        ylabel(900(200)1400) ymtick(##5)                ///
        legend(label(1 "最高价") label(2 "最低价"))      ///
        note("资料来源: Stata公司, SP500.dta")           ///
        caption("说明: 我做的第一幅Stata图形!")         ///
        saving(SP500.gph, replace)

(file SP500.gph saved)

. cap graph export SP500.png, replace
```

Basic Graph : Introduction

2. 二维图命令

▶ 实例 1



Basic Graph : Introduction

- 2. 二维图命令


- ▶ 实例 2

```
. twoway (function y=sin(x), range(-20 20) lw(*1.5)) ///  
         (function y=cos(x), range(-20 20) lw(*2.0)), ///  
         ytick(-2(0.5)2) ylabel(, angle(0))          ///  
         yline(0, lcolor(black*0.2) lpattern(dash))  ///  
         scheme(s1rcolor )                          ///  
         saving(trigono.gph, replace)  
  
. cap graph export trigono.png, replace
```

Basic Graph : Introduction

- 2. 二维图命令

- ▶ 实例 2



● 2. 二维图命令

- ▶ 图形无非是点、线（面）、文字等元素的组合
- ▶ 这些组合的整体“风格”构成了图类：单元图（，前的部分）
- ▶ 每种图形的具体特征由元素的特征决定：选项（，后的部分）
- ▶ 因此，选项的填写是 Stata 绘图的关键！

Subsection 2

Management

Basic Graph : Management

- 1. 保存

- ▶ 【方法一】 `-graph save-`

```
sysuse sp500, clear
twoway line high date
graph save fig1.gph, replace
```

- ▶ 【方法二】 `saving(filename, ...)`

```
twoway line high date, saving(fig1.gph, replace)
```

- ▶ 【方法三】 右击 —> `Save graph ...`

Basic Graph : Management

- 2. 导出

- ▶ 可选择的图形格式: .ps .eps .svg .wmf .emf .pdf .png .tif .gif .jpg ...

```
graph export fig1.wmf, replace
```

```
graph export "D:\Teaching\Stata\lab6\fig1.wmf", replace
```

^^I*调整图片的分辨率(适用于.png 和 .tif)

```
graph export fig1.tif, width(800) height(600) replace
```

Basic Graph : Management

- 3. 调入

```
graph use trigono.gph  
graph use trigono, scheme(s1mono)
```

- 4. 显示

```
graph display, scheme(sj)
```

Basic Graph : Management

- 5. 查询

```
graph dir
```

- 6. 删除

```
erase trigono.gph  
graph drop _all
```

Basic Graph : Management


● 7. 合并

```
. help graph combine
```

```
. sysuse sp500, clear  
(S&P 500)  
. twoway line high low date  
. graph save fig2.gph, replace  
(file fig2.gph saved)  
. twoway line high low date, scheme(s1mono)  
. graph save fig3.gph, replace  
(file fig3.gph saved)  
. graph combine fig2.gph fig3.gph , saving(sp500_c,replace)  
(file sp500_c.gph saved)  
  
. cap graph export sp500_c.png, replace
```

Basic Graph : Management

- 7. 合并




Basic Graph : Management

8. 显示模式

▶ e.g.

```
. use allstates.dta, clear  
. scatter propval100 rent700 ownhome, scheme(vg_outm)  
. cap graph export vg1.png, replace
```



Basic Graph : Management

8. 显示模式

▶ e.g.

```
. scatter propval100 rent700 ownhome, scheme(vg_outc)  
. cap graph export vg1.png, replace
```



- 8. 显示模式

- ▶ 其他模板

```
findit scheme
```

Subsection 3

title_options

Basic Graph : title_options


- 标题种类：主标题、副标题、注释、说明
- *title()*、*subtitle()*、*note()*、*caption()*

```
help title_options
```

Basic Graph : title_options

- 1. 主标题


```
. use allstates.dta, clear  
(Data on 50 States)  
. scatter propval100 ownhome, title("My title")  
. cap graph export ti_1.png, replace
```



Basic Graph : title_options

2. 副标题


```
. scatter propval100 ownhome, title("My title") subtitle("My subtitle")  
  
. cap graph export ti_2.png, replace
```



Basic Graph : title_options

3. 说明和注释


```
. scatter propval100 ownhome, title("My title") subtitle("My subtitle") ///  
caption("My caption") note("My note")  
  
. cap graph export ti_3.png, replace
```



Basic Graph : title_options

4. 标题的位置

```
. scatter propval100 ownhome, title("My title", position(6))  
  
. cap graph export ti_4.png, replace
```



Basic Graph : title_options

4. 标题的位置


```
. scatter propval100 ownhome, title("My title", position(1) ring(0))
```

```
. cap graph export ti_5.png, replace
```

* ring(0) = 绘图区内


* ring(k), k>0, 绘图区以外

* ring()的值越大, 距离绘图区越远



Basic Graph : title_options

```
. scatter propval100 ownhome, title("This is my" "title",    ///  
    box bexpand justification(left) span)                ///  
  
. cap graph export ti_6.png, replace  
  
^^I*box : add a box  
^^I*bexpand : fill the width of the plot region  
^^I*ustification(left) : left justify the text inside the box  
^^I*span : make the box span the entire width of the graph
```



Subsection 4

axis_options

- 1. 坐标轴刻度 (tick) 及刻度标签 (label)


```
help axis_label_options
```

- ▶ 主刻度及标签: *ylabel()*, *xlabel()*—major ticks plus labels
- ▶ 主刻度: *ytick()*, *xtick()*—major ticks only
- ▶ 子刻度及标签: *ymlabel()*, *xmlabel()*—minor ticks plus labels
- ▶ 子刻度: *ymtick()*, *xmtick()*—minor ticks only

Basic Graph : axis_options

- 1. 坐标轴刻度 (tick) 及刻度标签 (label)

- . twoway scatter propval100 faminc, xlabel(#10) ylabel(#10)
- . cap graph export ax_1.png, replace
- * ask for about 10 values to be labeled on each axis




Basic Graph : axis_options

1. 坐标轴刻度 (tick) 及刻度标签 (label)

```
twoway scatter propval100 faminc, ylabel(0(10)100)  
twoway scatter propval100 faminc, ylabel(10(20)90)  
twoway scatter propval100 faminc, ytick(#10)  
twoway scatter propval100 faminc, ymtick(##10)
```


```
. twoway scatter propval100 faminc, xlabel(,angle(45)) ylabel(,angle(-15))  
. cap graph export ax_2.png, replace
```



Basic Graph : axis_options

- 1. 坐标轴刻度 (tick) 及刻度标签 (label)

```
. twoway scatter propval100 region, xlabel(1 "NorthEast" 2 ///  
      "NorthCentral" 3 "Southern" 4 "Western")  
  
. cap graph export ax_3.png, replace
```




- 2. 坐标轴标题: `ytitle()` `xtitle()`

```
help axis_title_options
```

Basic Graph : axis_options

- 2. 坐标轴标题: ytitle() xtitle()

```
. twoway scatter ownhome propval100,          ///  
    ytitle("Percent of households that own their homes")  ///  
    xtitle("Percent of homes over $100,000", size(small) color(navy) place(right))  
  
. cap graph export ax_4.png, replace
```



- 3. 坐标结构: `yscale()` `xscale()`


```
help axis_scale_options
```

Basic Graph : axis_options

3. 坐标结构: yscale() xscale()

```
. twoway scatter educ popden, xscale(log)  
. cap graph export ax_5.png, replace
```

* the axis should be displayed on a log scale




Basic Graph : axis_options

3. 坐标结构: yscale() xscale()

- . twoway scatter educ popden, **yscale(noline) xscale(noline)**
- . cap graph export ax_6.png, replace

*不显示坐标轴




Basic Graph : axis_options

3. 坐标结构: yscale() xscale()

- . twoway scatter educ popden, **xscale(range(-1000 11000))**
- . cap graph export ax_7.png, replace

* 显示范围




Basic Graph : axis_options

3. 坐标结构: yscale() xscale()

```
. twoway scatter educ popden, xscale(lcolor(red) lwidth(vthick))
```

```
. cap graph export ax_8.png, replace
```

* 坐标轴线型



- 4. 双坐标系


```
help axis_choice_options
```

Basic Graph : axis_options

4. 双坐标系

* 共用 x 轴

```
. twoway (scatter faminc educ, yaxis(1)) ///  
        (scatter workers2 educ, yaxis(2))  
. cap graph export ax_9.png, replace
```




Basic Graph : axis_options

4. 双坐标系

```
. twoway (scatter faminc educ)          ///  
         (scatter workers2 educ, yaxis(2)), ///  
         ylabel(40(5)80, axis(2))
```

```
. cap graph export ax_10.png, replace
```

* Without the axis(2) option, Stata would assume that we are referring to the first axis



Subsection 5

legend_options


Basic Graph : legend_options

- 一张图中同时呈现多个序列，便会自动产生图例
- 对于变量而言，其默认图例是它的变量标签

```
help legend_options
```


Basic Graph : legend_options

```
. set scheme vg_s2c  
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban)  
. cap graph export le_1.png, replace
```




Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban), ///  
    legend(order(2 3 1))  
  
. cap graph export le_2.png, replace
```




Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban), ///  
    legend(label(1 "% Own home") label(2 "Lin.Fit") label(3 "Quad. Fit"))  
  
. cap graph export le_3.png, replace
```




Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban), ///  
    legend(order(1 "% Own Home" 2 "Linear" 3 "Quad") rows(1) position(12))  
  
. cap graph export le_4.png, replace
```




Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban), ///  
    legend(rows(1) title("Legend", color(red) size(huge) box bexpand))  
  
. cap graph export le_5.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban), ///  
    legend(size(large) color(maroon) fcolor(eggshell) box)  
  
. cap graph export le_6.png, replace
```



Subsection 6

region_options

Basic Graph : region_options

- Stata 图形的区域划分

```
help region_options
```

Basic Graph : region_options

- Stata 图形的区域划分


```
. sysuse auto, clear
(1978 Automobile Data)

. scatter mpg weight,                                     ///
  graphregion(fcolor(lavender*0.8))                    ///
  graphregion(ifcolor(eggshell))                       ///
  plotregion(fcolor(black*0.3))                        ///
  plotregion(ifcolor(white))                           ///
  title("全图内区 (inner graph region)")               ///
  xtitle("汽车重量")                                   ///
  ytitle("汽车里数")                                   ///
  note("(1) 灰色部分-- “绘图外区” (outer plot region)"  ///
        "(2) 紫色部分-- “全图外区” (outer graph region)"  ///
  caption("各区域的面积决定于显示模式(scheme)和内容的多少") ///
  text(30 3500 "绘图内区 (inner plot region)", color(blue) size(*1.5))
```

Basic Graph : region_options

- Stata 图形的区域划分

```
. cap graph export re_1.png, replace
```



Subsection 7

`added_line_options`

- 附加线

```
^^I    help added_line_options
```

```
^^I
```

- ▶ 语法

```
^^Itwoway ..., yline(数字, 子选项) xline(数字, 子选项)
```

```
^^I
```

- ▶ 数字: 控制附加线的位置
- ▶ 子选项: 控制附加线的类型、颜色、宽度等

Basic Graph : added_line_options

● 附加线

```
. sysuse sp500.dta,clear  
  
. line open date,                               ///  
    yline(1100,lwidth(*1.5) lpattern(shortdash_dot) lcolor(blue*0.6))  ///  
    yline(1313,lw(*2.5) lc(green*0.4))                                     ///  
    xline(15242,lw(*2) lc(pink*0.4) lp(longdash))  
  
. cap graph export al_1.png, replace
```



Subsection 8

added_text_options

- 文字与文本框

help textbox_options

- ▶ 凡是出现文字的地方都可以做下面的设定
- ▶ 文字和文本框的整体风格：标题、副标题、文本、小号
- ▶ 文本框相关设定：文本框颜色、背景、与文字的边距等
- ▶ 文字相关的设定：大小、颜色、位置、行距等

Basic Graph : added_text_options

- 文字与文本框

```
. line open date, text(1324.83 15117 "一个波峰",color(blue) size(*1.6))
```

```
. cap graph export at_1.png, replace
```



Subsection 9

`marker_options` & `marker_label_options`

Basic Graph : marker_options & marker_label_options


- 图标和图标的标签

```
help marker_options  
help marker_label_options
```

Basic Graph : marker_options & marker_label_options

- 图标和图标的标签


```
. use allstates.dta, clear  
(Data on 50 States)  
. twoway scatter ownhome borninstate, msymbol(X) msize(small)  
  
. cap graph export ma_1.png, replace
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签


```
. twoway scatter ownhome borninstate [aweight=propval100], ///  
    msymbol(oh) msize(small)  
  
. cap graph export ma_2.png, replace
```



Basic Graph : marker_options & marker_label_options

● 图标和图标的标签

```
. twoway scatter ownhome borninstate [aweight=propval100], ///  
    msymbol(oh) msize(large) mlabel(stateab)  
  
. cap graph export ma_3.png, replace  
* the weights no longer affect the size of the markers
```




Basic Graph : marker_options & marker_label_options

● 图标和图标的标签

```
. twoway (scatter ownhome borninstate [aweight=propval100], msymbol(oh) msize(large)) /
        (scatter ownhome borninstate, mlabel(stateab) msymbol(i) mlabpos(center))

. cap graph export ma_4.png, replace


/* To solve:overlying a scatterplot that has the symbols weighted by propval100
with a scatterplot that shows just the marker labels.*/
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签


- . twoway scatter ownhome borninstate, mfcolor(ltblue) mlcolor(navy) mlwidth(medthick)
- . cap graph export ma_5.png, replace



Basic Graph : marker_options & marker_label_options

● 图标和图标的标签

- `tway scatter ownhome borninstate, ///`
`mlabel(stateab) mlabpos(12) mlabsize(vsmall) mlabangle(45) mlabcolor(red)`
- `cap graph export ma_6.png, replace`



Subsection 10

by_option

Basic Graph : by_option


- 分组绘图

```
help by_option
```

Basic Graph : by_option

- 分组绘图


```
. use allstates.dta, clear  
(Data on 50 States)  
. twoway scatter ownhome borninstate, by(north)  
. cap graph export by_1.png, replace
```



Basic Graph : by_option

● 分组绘图


- . twoway scatter ownhome borninstate, by(north, total)
- . cap graph export by_2.png, replace



Basic Graph : by_option

- 分组绘图


- `twoway scatter ownhome borninstate, by(north, total holes(2))`
- `cap graph export by_3.png, replace`



Basic Graph : by_option

● 分组绘图


- . twoway scatter ownhome borninstate, by(north, total rows(1))
- . cap graph export by_4.png, replace



Basic Graph : by_option

- 分组绘图

- `twoway scatter ownhome borninstate, by(north, total iscale(*1.5))`
- `cap graph export by_5.png, replace`




Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, rescale)  
. cap graph export by_6.png, replace
```


* scales both the variable and variable differently across the by-groups. Both axes are separately rescaled.



Basic Graph : by_option

- 分组绘图

- `twoway scatter ownhome borninstate, by(north, title("By title")) title("Regular title")`
- `cap graph export by_7.png, replace`



Basic Graph : by_option

- 分组绘图

```
. twoway scatter (borninstate propval100 ownhome), ///  
> by(nsw,legend(position(12))) legend(label(1 "Born in state") label(2 "% > 100K"))  
. cap graph export by_8.png, replace
```



Subsection 11


linear prediction plots

Basic Graph : linear prediction plots

```
help twoway lfit  
help twoway qfit
```

● 简单示例

```
. scatter ownhome pcturban80 || lfit ownhome pcturban80, by(north, total row(1))  
. cap graph export lp_1.png, replace
```




Basic Graph : linear prediction plots

```
help twoway lfitci  
help twoway qfitci
```

● 附加置信区间

* 线性拟合的置信区间

```
. twoway (lfitci ownhome pcturban80, stdf) (scatter ownhome pcturban80)  
. cap graph export lp_2.png, replace
```




Basic Graph : linear prediction plots

* 非线性拟合


```
. twoway (qfitci ownhome pcturban80, stdf) (scatter ownhome pcturban80)
```

```
. cap graph export lp_3.png, replace
```



Basic Graph : linear prediction plots

```
. twoway (lfitci ownhome pcturban80, //  
*ciplot(rline) lcolor(green) lpattern(dash) lwidth(thick)) //  
  (scatter ownhome pcturban80)  
. cap graph export lp_4.png, replace
```



Basic Graph : linear prediction plots

```
. twoway (lfitci ownhome pcturban80, stdf level(99) color(eggshell)) ///  
        (lfitci ownhome pcturban80, stdf level(90) color(lavender)) ///  
        (scatter ownhome pcturban80, mcolor(black))  
  
. cap graph export lp_5.png, replace
```

